


Giochiamo con lo Spirografo

Lo spirografo è un gioco matematico per bambini, consistente in un insieme di rotelle dentate che con l'ausilio di penne colorate permettono di ricavare interessanti disegni geometrici. I principi matematici che stanno dietro allo spirografo sono molto semplici e costruirne uno virtuale è un facile esercizio.

Introduzione

Uno spirografo (Figura 1) consiste in una piastra forata e un insieme di rotelline, di solito costruite in plastica trasparente. Sia i fori all'interno della piastra che le rotelline sono munite di dentini, tutti compatibili tra loro, in modo che qualunque rotellina può ruotare senza slittare all'interno di qualunque foro. Il gioco consiste nel puntare un matita o un penna colorate in uno dei buchi all'interno delle rotelle e far girare le rotelle disegnando in tal modo un complicata linea curva.



L'equazione dello Spirografo

Consideriamo un foro con centro in $(0,0)$ e raggio m e una rotella con raggio n . Se m ed n sono entrambi interi si hanno due vantaggi, uno fisico e uno matematico:

- è possibile costruire un insieme di denti che permette la rotazione senza scorrimento
- dopo un certo numero di giri si torna allo stato iniziale.

Per semplificare i calcoli rappresentiamo i punti nel piano di Gauss invece che nell'ordinario piano $x-y$. Questo trucco permette di usare gli esponenziali complessi che (ad onta del nome) sono il modo più semplice per trattare i vettori rotanti.

Supponendo unitaria la velocità angolare del centro della rotella, l'equazione del moto di quest'ultimo è quindi:

$$(m-n) e^{2\pi j t}$$

dove j rappresenta l'unità immaginaria. Se il punto bianco è posto ad una distanza λn dal centro della rotella, considerando che la velocità angolare della rotella è m/n e la sua rotazione è nel senso inverso l'equazione del moto del punto bianco è

$$(m-n) e^{2\pi j t} + \lambda n e^{-2\pi j t m/n}$$

Il campo di variazione di t è tra 0 e $2\pi T$. Il valore di T per cui la figura si richiude esattamente è il minimo intero tale che anche $T m/n$ è intero. Infatti la rotella piccola ad ogni giro del suo centro fa m/n giri e solo se $T m/n$ entrambe le rotelle tornano nello stato iniziale.

Con questo abbiamo già finito la parte matematica del problema.

Implementazione in Mathematica

Spirografo

Il nostro spirografo virtuale si realizza con due sole funzioni. La prima funzione traccia il grafico nel piano di Gauss di una funzione complessa.

```
In[1]:=
GaussPlot[f_, t_, T_, opts___] :=
```

ParametricPlot[{Re[f], Im[f]}, {t, 0, T}, opts];

Si noti che i tre *underscore* (_) dopo **opts** stanno a significare che il parametro **opts** (che rappresenta le opzioni del grafico) può essere assente o consistere in un numero arbitrario di argomenti.


La seconda funzione disegna una curva del colore **col** con **m** denti nel foro, **n** denti nella rotella e con la penna puntata ad una frazione λ del raggio della rotella.

```
In[2]:=
Spiro[m_, n_, λ_, col_] := (
  T = Denominator[m/n];
  GaussPlot[(m-n)Exp[I t]+λ n Exp[-I t
m/n], t, 2π T,
  PlotStyle -> col,
  PlotPoints -> 50 T,
  AspectRatio -> 1,
  PlotRange->{{-1.1m, 1.1m}, {-1.1m, 1.1m}});
```

Il numero dei punti del plottaggio è reso proporzionale al numero **T** dei giri del centro della rotella.

Ecco subito un esempio semplice (Figura 2)

```
In[3]:=
Spiro[36, 15, 0.4, Red];
```



Sovrapposizione


Nel gioco reale il bello sta nel sovrapporre più curve di colori diversi. Si può estendere a questo scopo la funzione **Spiro**.

```
In[4]:=
Spiro[v_List] := Show[
  Block[{$DisplayFunction = Identity},
    Spiro[Sequence@@#]&/@v]]
```

Spieghiamo bene questa tecnica perché può essere utile anche in generale. **f[Sequence@@#]&** esegue **f** su una lista "togliendo le parentesi graffe", ovvero **f[Sequence@@#]&** applicato ad **{a,b,c}** esegue **f[a,b,c]**. Questa funzione pura viene applicata alla lista delle liste degli argomenti e racchiusa in **Block[{\$DisplayFunction=Identity},...]** per inibire le visualizzazioni dei grafici intermedi. Al tutto viene applicato **Show** per visualizzare un risultato solo.

Un esempio è presentato in (Figura 3).

```
In[5]:=
Spiro[{
  {36, 17, 0.7, Green},
  {23, 15, 0.4, Red},
  {36, 14, 0.5, Blue}}];
```



Rotelle

Per gli scettici vediamo di mostrare come le figure ottenute siano proprio il risultato della rotazione delle rotelle.

Facciamo un programma che anima le rotelle rotanti disegnando anche i raggi e i punti bianchi. La funzione **pl[t]** rappresenta il fotogramma al tempo **t**.

```
In[6]:=
pl[t_] := Graphics[{
  Green,
  Rectangle[{-1.1m, -1.1m}, {1.1m, 1.1m}],
  White,
  Disk[{0, 0}, m],
  Red,
  Disk[(m - n){Cos[t], Sin[t]}, n],
  Black, Thickness[0.002], Line[{{0, 0},
  (m - n){Cos[t], Sin[t]},
  (m - n){Cos[t], Sin[t]} + λ n{Cos[-t m/n],
  Sin[-t m/n]}],
```

```
White,
Disk[(m - n){Cos[t],Sin[t]} +
  λ n{Cos[-t m/n],Sin[-t m/n]}, 0.02m],
Black}}];
```

Si disegnano molti fotogrammi raccogliendoli in una lista e sovrapponendo ogni volta la stessa curva disegnata in blu.

```
In[7]:=
m = 10;
n = 4;
λ = 0.6;
δ = π/9;
sp = Spiro[m, n, λ, Blue];
tt = Table[Show[pl[t], sp,
  AspectRatio -> 1,
  PlotRange -> {{-1.1m,1.1m}, {-1.1m,1.1m}}],
  {t, δ, 4π, δ}]
```

Un fotogramma è mostrato in (Figura 4).



I primi nove fotogrammi della intera sequenza sono mostrati invece nel seguente GraphicsArray (Figura 5).

```
In[8]:=
Show[GraphicsArray[Partition[Take[tt, 9],
3]]];
```

Il bambino virtuale

In genere quando si regala uno spirografo ad un bambino questi si mette ad usarlo ripetutamente con colori diversi fino riempire il foglio di bei disegni geometrici.

Vediamo di fare altrettanto in modo virtuale. Innanzitutto modifichiamo Spiro aggiungendo un quinto argomento che determina la posizione del centro del disegno sul piano cartesiano (la conversione al piano di Gauss viene effettuata all'interno moltiplicando per {1, I}).

```
In[1]:=
Spiro[m_, n_, λ_, col_, p_] := (
  T = Denominator[m/n];
  GaussPlot[p.{1, I}+
    (m-n)Exp[I t]+
    λ n Exp[-I t m/n],
    t, 2π T,
    PlotStyle -> col,
    PlotPoints -> 50T])
```

Poi si tratta di determinare l'area interessata da tutti i disegni e scalare il grafico in modo da avere una rappresentazione non deformata. PlotRange[Show[... , PlotRange->All]] rende il valore corretto del PlotRange (nella forma {{x0,x1},{y0,y1}}) per mostrare tutto quello che è nel grafico. Da questo valore si può calcolare il valore dell'AspectRatio che preserva le forme. attraverso l'espressione (y1-y0)/(x1-x0)

```
In[2]:=
Spiro[v_List] := (
  Block[{$DisplayFunction = Identity},
    {{x0,x1},{y0,y1}} = PlotRange[gg =
      Show[Spiro[Sequence@@#]&/@v,
        PlotRange -> All]]];
  Show[gg,
    PlotRange->{{x0,x1},{y0,y1}},
    Axes -> False,
    AspectRatio ->
      (y1-y0)/(x1-x0))
```





Figura 5

Ecco un disegno con tre spirali.

```
In[3]:=
Spiro[ {
  {36, 17, 0.7, Green, {0, 0}},
  {23, 15, 0.4, Red, {10, 10}},
  {36, 14, 0.5, Blue, {20, 10}} ]];
```

(Figura 6).




Adesso generiamo un po' di grafici con parametri casuali.

```
In[4]:=
Do[ Spiro[ Table[
  {m = Random[Integer, {23, 47}],
  m - Random[Integer, {3, 20}],
  Random[], Hue[k/5],
  {50Random[], 30Random[]}], {k, 5}], {10}]
```

Ne ho generati 10 e ho scelto quelli che piacevano di più a mia figlia. (Figura 7-8).

Figura 8



MC

Bibliografia

Molte sono le variazioni sul tema. Noi abbiamo cercato di riprodurre per quanto possibile il gioco fisico che viene acquistato. Per ulteriori interessanti spunti si possono vedere i lavori citati di seguito.

Stan Wagon, **Mathematica in Action**, Capitolo 2. Ed. Italiana: **Guida a Mathematica** (McGraw Hill, 1995).

Spirograph.nb pacchetto dimostrativo nella distribuzione di **Mathematica**.

Un'ultima segnalazione al volo. Vale la pena di navigare alla URL <http://mathworld.wolfram.com> dove si trova una specie di enciclopedia della matematica fatta con **Mathematica** (prima o poi ci dedico un articolo).

Figura 7

